SOLUCION

FACULTAD DE INGENIERÍA
ECUACIONES DIFERENCIALES
PRIMER EXAMEN PARCIAL

2018 SEPTIEMBRE 27

```
> restart:
```

1) **(40/100 puntos)** DADAS LA SIGUIENTE ECUACIÓN DIFERENCIAL ORDINARIA Y SU SOLUCIÓN GENERAL

$$ecuacion\_diferencial := y(x) = x\left(\frac{d}{dx}y(x)\right) + \left(\frac{d}{dx}y(x)\right)^2 \quad \textbf{(1)}$$

$$solucion\_general := y(x) = x\,\_C1 + \_C1^2$$

E INDIQUE CUÁLES DE LAS SIGUIENTES FUNCIONES SON SOLUCIÓN Y DE QUÉ TIPO **(general, particular o singular)** Y CUÁLES NO LO SON, ARGUMENTANDO CADA RESULTADO

$$funcion\_1 := y(x) = -\frac{x}{4} + \frac{1}{16} \quad \textbf{(2)}$$

$$funcion\_2 := y(x) = \frac{x}{4} + \frac{1}{16}$$

$$funcion\_3 := y(x) = \frac{x^2}{16}$$

$$funcion\_4 := y(x) = \frac{x^2}{4}$$

$$funcion\_5 := y(x) = -\frac{x^2}{4}$$

```
> restart:
```

RESPUESTA 1)

```
> ecuacion_diferencial := y(x) = x*diff(y(x),x)+diff(y(x),x)^2;
```

$$ecuacion\_diferencial := y(x) = x\left(\frac{d}{dx}y(x)\right) + \left(\frac{d}{dx}y(x)\right)^2 \quad \textbf{(3)}$$

```
> solucion_general := y(x) = x*_C1+_C1^2;
```

$$solucion\_general := y(x) = x\,\_C1 + \_C1^2 \quad \textbf{(4)}$$

```
> funcion_1 := y(x) = -1/4*x+1/16;funcion_2 := y(x) = 1/4*x+1/16;
  funcion_3 := y(x) = 1/16*x^2;funcion_4 := y(x) = 1/4*x^2;
  funcion_5 := y(x) = -1/4*x^2;
```

$$funcion\_1 := y(x) = -\frac{x}{4} + \frac{1}{16} \quad \textbf{(5)}$$

$$funcion\_2 := y(x) = \frac{x}{4} + \frac{1}{16}$$

$$function\_3 := y(x) = \frac{x^2}{16}$$

$$function\_4 := y(x) = \frac{x^2}{4}$$

$$function\_5 := y(x) = -\frac{x^2}{4}$$

```
> comprobacion_0:=simplify(eval(subs(y(x)=rhs(solucion_general),lhs
  (ecuacion_diferencial)-rhs(ecuacion_diferencial)=0)));
```

$$comprobacion\_0 := 0 = 0 \qquad \textbf{(6)}$$

la solución_general satisface la ecuación

```
> comprobacion_1:=simplify(eval(subs(y(x)=rhs(funcion_1),lhs
  (ecuacion_diferencial)-rhs(ecuacion_diferencial)=0)));
```

$$comprobacion\_1 := 0 = 0 \qquad \textbf{(7)}$$

```
> parametro_1:=solve(rhs(solucion_general)=rhs(funcion_1),_C1);
```

$$parametro\_1 := -\frac{1}{4}, -x + \frac{1}{4} \qquad \textbf{(8)}$$

la funcion_1 satisface la ecuación diferencial es solucion y como existe _C1 = -1/4 y este es un valor real entonces la función_1 es una solución particular

```
> comprobacion_2:=simplify(eval(subs(y(x)=rhs(funcion_2),lhs
  (ecuacion_diferencial)-rhs(ecuacion_diferencial)=0)));
```

$$comprobacion\_2 := 0 = 0 \qquad \textbf{(9)}$$

```
> parametro_2:=solve(rhs(solucion_general)=rhs(funcion_2),_C1);
```

$$parametro\_2 := \frac{1}{4}, -x - \frac{1}{4} \qquad \textbf{(10)}$$

como la funcion_2 satisface la ecuación diferencial es solucion y como existe _C1 = 1/4 y este es un valor real entonces la función_2 es una solución particular

```
> comprobacion_3:=simplify(eval(subs(y(x)=rhs(funcion_3),lhs
  (ecuacion_diferencial)-rhs(ecuacion_diferencial)=0)));
```

$$comprobacion\_3 := -\frac{5\,x^2}{64} = 0 \qquad \textbf{(11)}$$

como la funcion_3 no satisface la ecuación diferencial por lo tanto no es solución

```
> comprobacion_4:=simplify(eval(subs(y(x)=rhs(funcion_4),lhs
  (ecuacion_diferencial)-rhs(ecuacion_diferencial)=0)));
```

$$comprobacion\_4 := -\frac{x^2}{2} = 0 \qquad \textbf{(12)}$$

como la funcion_4 no satisface la ecuación diferencial por lo tanto no es solución

```
> comprobacion_5:=simplify(eval(subs(y(x)=rhs(funcion_5),lhs
  (ecuacion_diferencial)-rhs(ecuacion_diferencial)=0)));
```

$$comprobacion\_5 := 0 = 0 \qquad \textbf{(13)}$$

```
> parametro_5:=solve(rhs(solucion_general)=rhs(funcion_5),_C1);
```

$$parametro\_5 := -\frac{x}{2}, -\frac{x}{2} \qquad (14)$$

como la funcion_5 satisface la ecuación diferencial es solución pero como _C1 no toma valor real alguno entonces la funcion_5 es una solución singular

FIN RESPUESTA 1)

```
> restart:
```

2) **(30/100 puntos)** OBTENGA LA SOLUCIÓN PARTICULAR DE LA SIGUIENTE ECUACIÓN DIFERENCIAL CON LA CONDICIÓN INICIAL DADA - UTILIZANDO EXCLUSIVAMENTE EL MÉTODO DE FACTOR INTEGRANTE - **(no se puede utilizar dsolve ni exactsol; pero sí se puede utilizar intfactor)**

$$ecuacion\_diferencial := 2\,x\,y(x)^2 - 3\,y(x)^3 + \left(7 - 3\,x\,y(x)^2\right)\left(\frac{d}{dx}\,y(x)\right) = 0 \qquad (15)$$

$$condicion := y(1) = 2$$

```
> restart:
```

RESPUESTA 2)

```
> ecuacion_diferencial := 2*x*y(x)^2-3*y(x)^3+(7-3*x*y(x)^2)*diff(y
  (x),x) = 0;condicion := y(1) = 2;
```

$$ecuacion\_diferencial := 2\,x\,y(x)^2 - 3\,y(x)^3 + \left(7 - 3\,x\,y(x)^2\right)\left(\frac{d}{dx}\,y(x)\right) = 0 \qquad (16)$$

$$condicion := y(1) = 2$$

```
> M(x,y):=2*x*y^2-3*y^3;
```

$$M(x,y) := 2\,x\,y^2 - 3\,y^3 \qquad (17)$$

```
> N(x,y):=7-3*x*y^2;
```

$$N(x,y) := 7 - 3\,x\,y^2 \qquad (18)$$

```
> comprobacion_1:=simplify(diff(M(x,y),y)-diff(N(x,y),x)=0);
```

$$comprobacion\_1 := 4\,x\,y - 6\,y^2 = 0 \qquad (19)$$

dado que no cumple con el teorema de Schwarz entonces la ecuación diferencial es NO_EXACTA

```
> with(DEtools):
```

```
> intfactor(ecuacion_diferencial);
```

$$\frac{1}{y(x)^2} \qquad (20)$$

```
> FI:=1/(y^2);
```

$$FI := \frac{1}{y^2} \qquad (21)$$

```
> MM(x,y):=expand(FI*M(x,y));
```

$$MM(x,y) := 2\,x - 3\,y \qquad (22)$$

```
> NN(x,y):=expand(FI*N(x,y));
```

$$NN(x,y) := \frac{7}{y^2} - 3\,x \quad \textbf{(23)}$$

```
> comprobacion_2:=diff(MM(x,y),y)-diff(NN(x,y),x)=0;
```

$$comprobacion\_2 := 0 = 0 \quad \textbf{(24)}$$

```
> solucion_general:=int(MM(x,y),x)+int((NN(x,y)-diff(int(MM(x,y),
  x),y)),y)=_C1;
```

$$solucion\_general := x^2 - 3\,x\,y - \frac{7}{y} = \_C1 \quad \textbf{(25)}$$

```
> parametro:=isolate(subs(x=1,y=2,solucion_general),_C1);
```

$$parametro := \_C1 = -\frac{17}{2} \quad \textbf{(26)}$$

```
> solucion_particular:=subs(_C1=rhs(parametro),solucion_general);
```

$$solucion\_particular := x^2 - 3\,x\,y - \frac{7}{y} = -\frac{17}{2} \quad \textbf{(27)}$$

COMPROBACION

```
> solucion:=x^2-3*x*y(x)-7/y(x) = -17/2;
```

$$solucion := x^2 - 3\,x\,y(x) - \frac{7}{y(x)} = -\frac{17}{2} \quad \textbf{(28)}$$

```
> ecuacion_1:=simplify(isolate(diff(solucion,x),diff(y(x),x)));
```

$$ecuacion\_1 := \frac{\mathrm{d}}{\mathrm{d}x}\,y(x) = \frac{(2\,x - 3\,y(x))\,y(x)^2}{3\,x\,y(x)^2 - 7} \quad \textbf{(29)}$$

```
> ecuacion_diferencial;
```

$$2\,x\,y(x)^2 - 3\,y(x)^3 + \left(7 - 3\,x\,y(x)^2\right)\left(\frac{\mathrm{d}}{\mathrm{d}x}\,y(x)\right) = 0 \quad \textbf{(30)}$$

```
> ecuacion_2:=isolate(ecuacion_diferencial,diff(y(x),x));
```

$$ecuacion\_2 := \frac{\mathrm{d}}{\mathrm{d}x}\,y(x) = \frac{-2\,x\,y(x)^2 + 3\,y(x)^3}{7 - 3\,x\,y(x)^2} \quad \textbf{(31)}$$

```
> comprobacion_3:=simplify(rhs(ecuacion_1)-rhs(ecuacion_2))=0;
```

$$comprobacion\_3 := 0 = 0 \quad \textbf{(32)}$$

como la ecuacion_1 que se obtuvo de la solución particular obtenida por el método de FACTOR INTEGRANTE es igual a la ecuacion_2 obtenida a partir de la ecuación diferencial original entonces se comprueba que la solución particular obtenida satisface la ecuación diferencial original.

FIN RESPUESTA 2)

```
> restart:
```

3) DADA LA SIGUIENTE ECUACIÓN DIFERENCIAL DE COEFICIENTES HOMOGENEOS (**no se puede utilizar dsolve)**

$$ecuacion\_diferencial := x - y(x)\cos\left(\frac{y(x)}{x}\right) + x\cos\left(\frac{y(x)}{x}\right)\left(\frac{d}{dx}y(x)\right) = 0 \quad (33)$$

a) (**15/100 puntos**) OBTENGA LA SOLUCIÓN GENERAL

b) (**15/100 puntos**) GRAFIQUE LA SOLUCIÓN PARTICULAR QUE SATISFACE LA

$$condicion\_inicial := y(\pi) = 2 \quad (34)$$

EN EL

$$intervalo := x = \pi..3\,\pi \quad (35)$$

RESPUESTA 3a)

```
> ecuacion_diferencial := x-y(x)*cos(y(x)/x)+x*cos(y(x)/x)*diff(y
  (x),x) = 0;
```

$$ecuacion\_diferencial := x - y(x)\cos\left(\frac{y(x)}{x}\right) + x\cos\left(\frac{y(x)}{x}\right)\left(\frac{d}{dx}y(x)\right) = 0 \quad (36)$$

como es de COEFICIENTES HOMOGÉNEOS entonces con la substitución y(x) = u(x)*x la transformará en una de VARIABLES SEPARABLES

```
> ecuacion_transformada:=simplify(eval(subs(y(x)=u(x)*x,
  ecuacion_diferencial)));
```

$$ecuacion\_transformada := x\left(1 + \cos(u(x))\left(\frac{d}{dx}u(x)\right)x\right) = 0 \quad (37)$$

```
> ecuacion_separable:=isolate(ecuacion_transformada,diff(u(x),x))*
  cos(u(x));
```

$$ecuacion\_separable := \cos(u(x))\left(\frac{d}{dx}u(x)\right) = -\frac{1}{x} \quad (38)$$

```
> ecuacion_separada:=lhs(ecuacion_separable)-rhs
  (ecuacion_separable)=0;
```

$$ecuacion\_separada := \cos(u(x))\left(\frac{d}{dx}u(x)\right) + \frac{1}{x} = 0 \quad (39)$$

```
> P(u):=cos(u);
```

$$P(u) := \cos(u) \quad (40)$$

```
> Q(x):=1/x;
```

$$Q(x) := \frac{1}{x} \quad (41)$$

```
> solucion_intermedia:=int(P(u),u)+int(Q(x),x)=_C1;
```

$$solucion\_intermedia := \sin(u) + \ln(x) = \_C1 \quad (42)$$

```
> solucion_general:=subs(u=y/x,solucion_intermedia);
```

(43)

$$solucion\_general := \sin\left(\frac{y}{x}\right) + \ln(x) = \_C1 \tag{43}$$

RESPUESTA 4b)

```
> condicion_inicial := y(Pi) = 2;intervalo := x = Pi .. 3*Pi;
```

$$condicion\_inicial := y(\pi) = 2 \tag{44}$$

$$intervalo := x = \pi..3\,\pi$$

```
> parametro:=isolate(subs(x=Pi,y=2,solucion_general),_C1);
```

$$parametro := \_C1 = \sin\left(\frac{2}{\pi}\right) + \ln(\pi) \tag{45}$$

```
> solucion_particular:=subs(_C1=rhs(parametro),solucion_general);
```

$$solucion\_particular := \sin\left(\frac{y}{x}\right) + \ln(x) = \sin\left(\frac{2}{\pi}\right) + \ln(\pi) \tag{46}$$

```
> solucion:=isolate(solucion_particular,y);
```

$$solucion := y = \arcsin\left(\sin\left(\frac{2}{\pi}\right) + \ln(\pi) - \ln(x)\right) x \tag{47}$$

```
> plot(rhs(solucion),intervalo);
```




```
> with(plots):
```

Warning, the name changecoords has been redefined

```
> implicitplot(solucion_particular,x=Pi..3*Pi,y=-6..3);
```



FIN RESPUESTAS 4)

```
> restart:
```

FIN DEL EXAMEN